
	1st Modified Section


6.3
Operation parameter mapping

…

Table 2: Mapping from IS getMoAttributes parameters to SS equivalents

	IS Operation parameter
	SS Method parameter
	Qualifier

	invokeIdentifier
	- (No equivalence)
	-

	invokeIdentifierOut
	Return value of type BasicCmInformationIterator
	M

	baseObjectInstance
	DN base_object
	M

	scope
	SearchControl search_control (SearchControl.type and SearchControl.level)
	M

	filter
	SearchControl search_control (SearchControl.filter)
	M

	attributeListIn
	AttributeNameSet requested_attributes
	M

	managerReference
	AlarmIRPConstDefs::ManagerReference
	O

	managedObjectClass
managedObjectInstance
attributeListOut
	Return value of type BasicCmInformationIterator - parameter out ResultSet fetched_elements of method next_basic_cm_informations
	M

	status
	Exceptions:

FindManagedObjects,
ManagedGenericIRPSystem::InvalidParameter,
UndefinedMOException,
IllegalDNFormatException,
UndefinedScopeException,
IllegalScopeTypeException,
IllegalScopeLevelException,
IllegalFilterFormatException,
FilterComplexityLimit
	M


…
[Rest in this section unchanged]

	Next Modified Section


A.1
IDL specification (file name "BasicCMIRPConstDefs.idl")

//File: BasicCMIRPConstDefs.idl

#ifndef _BASIC_CM_IRP_CONST_DEFS_IDL_

#define _BASIC_CM_IRP_CONST_DEFS_IDL_

// This statement must appear after all include statements

#pragma prefix "3gppsa5.org"

/* ## Module: BasicCMIRPConstDefs

This module contains commonly used definitions for BasicCMIRP.

================================================================

*/

module BasicCMIRPConstDefs

{

   /**

    * Defines the name of a Managed Object Class

    */

   typedef string MOClass;

   /**

    * The format of Distinguished Name (DN) is specified in 3GPP TS 32.300

    * "Name Conventions for Managed Objects".

    */

   typedef string DN;

   /**

    * Defines the name of an attribute of a Managed Object

    */

   typedef string MOAttributeName;

   /**

    * Defines the value of an attribute of a Managed Object in form of a CORBA

    * Any. Apart from basic datatypes already defined in CORBA, the allowed

    * attribute value types are defined in the AttributeTypes module.

    */

   typedef any MOAttributeValue;
   /**

    *

    * In this version the only allowed filter value is "TRUE" i.e. a filter that

    * matches everything.

    */

   typedef string Filter;

   /**

    * ResultContents is used to tell how much information to get back

    * from the find_managed_objects operation.

    *

    * NAMES: Used to get only Distinguished Name

    *        for MOs.

    *        The name contains both the MO class

    *        and the names of all superior objects in the naming

    *        tree.

    *

    * NAMES_AND_ATTRIBUTES: Used to get both NAMES plus

    *    MO attributes (all or selected).

    */

   enum ResultContents

   {

      NAMES,

      NAMES_AND_ATTRIBUTES

   };

   /**

    * ScopeType defines the kind of scope to use in a search

    * together with SearchControl.level, in a SearchControl value.

    *

    * SearchControl.level is always >= 0. If a level is bigger than the

    * depth of the tree there will be no exceptions thrown.

    * BASE_ONLY: level ignored, just return the base object.

    * BASE_NTH_LEVEL: return all subordinate objects that are on "level"

    * distance from the base object, where 0 is the base object.

    * BASE_SUBTREE: return the base object and all of its subordinates

    * down to and including the nth level.

    * BASE_ALL: level ignored, return the base object and all of it's

    * subordinates.

    */

   enum ScopeType

   {

      BASE_ONLY,

      BASE_NTH_LEVEL,

      BASE_SUBTREE,

      BASE_ALL

   };

   /**

    * SearchControl controls the find_managed_object search,

    * and contains:

    * the type of scope ("type" field),

    * the level of scope ("level" field), level 0 means the "baseObject",

    *    level 1 means baseobject including its sub-ordinates etc..

    * the filter ("filter" field),

    * the result type ("contents" field).

    * The type, level and contents fields are all mandatory.

    * The filter field contains the filter expression.

    *  The string "TRUE" indicates "no filter",

    * i.e. a filter that matches everything.

    */

   struct SearchControl

   {

      ScopeType type;

      unsigned long level;

      Filter filter_;

      ResultContents contents;

   };

   /**

    * Represents an attribute: "name" is the attribute name

    * and "value" is the attribute value.

    */

   struct MOAttribute

   {

      MOAttributeName name;

      MOAttributeValue value;

   };

   typedef sequence <MOAttribute> MOAttributeSet;

   struct Result

   {

      DN mo;

      MOAttributeSet attributes;

   };

   typedef sequence <Result> ResultSet;

   /**

    * AttributeErrorCategory defines the categories of errors, related to

    * attributes, that can occur during creation or modification of MOs.

    *

    * NO_SUCH_ATTRIBUTE: The specified attribute does not exist.

    * INVALID_ATTRIBUTE_VALUE: The specified attribute value is not valid.

    * MISSING_ATTRIBUTE_VALUE: An attribute value is required but none was

    *   provided and no default value is defined for the attribute.

    * INVALID_MODIFY_OPERATOR: The specified modify operator is not valid

    *   (e.g. operator ADD_VALUES applied to a non multi-valued attribute

    *   or operator SET_TO_DEFAULT applied where no default value is defined).

    * MODIFY_NOT_ALLOWED: The modification of the attribute is not allowed.

    * MODIFY_FAILED: The modification failed because of an unspecified reason.

    */

   enum AttributeErrorCategory

   {

      NO_SUCH_ATTRIBUTE,

      INVALID_ATTRIBUTE_VALUE,

      MISSING_ATTRIBUTE_VALUE,

      INVALID_MODIFY_OPERATOR,

      MODIFY_NOT_ALLOWED,

      MODIFY_FAILED

   };

   /**

    * DeleteErrorCategory defines the categories of errors that can occur

    * during deletion of MOs.

    *

    * SUBORDINATE_OBJECT: The MO cannot be deleted due to subordinate MOs.

    * DELETE_NOT_ALLOWED: The deletion of the MO is not allowed.

    * DELETE_FAILED: The deletion failed because of an unspecified reason.

    */

   enum DeleteErrorCategory

   {

      SUBORDINATE_OBJECT,

      DELETE_NOT_ALLOWED,

      DELETE_FAILED

   };

   /**

    * AttributeError represents an error, related to an attribute, that occured

    * during creation or modification of MOs.

    * It contains:

    * - the name of the indicted attribute ("name" field),

    * - the category of the error ("error" field),

    * - optionally, the indicted attribute value ("value" field),

    * - optionally, additional details on the error ("reason" field).

    */

   struct AttributeError

   {

      MOAttributeName name;

      AttributeErrorCategory error;

      MOAttributeValue value;

      string reason;

   };

   typedef sequence <AttributeError> AttributeErrorSeq;

   /**

    * DeleteError represents an error that occured during deletion of MOs.

    * It contains:

    * - the distinguished name of the indicted MO ("object_name" field),

    * - the category of the error ("error" field),

    * - optionally, additional details on the error ("reason" field).

    */

   struct DeleteError

   {

      DN object_name;

      DeleteErrorCategory error;

      string reason;

   };

   typedef sequence <DeleteError> DeleteErrorSeq;

   /**

    * ModifyAttributeErrors represents errors that occured during

    * modification of attributes of a MO.

    * It contains:

    * - the distinguished name of the indicted MO ("object_name" field),

    * - a sequence containing the attribute errors ("errors" field).

    */

   struct ModifyAttributeErrors

   {

      DN object_name;

      AttributeErrorSeq errors;

   };

   typedef sequence <ModifyAttributeErrors> ModifyAttributeErrorsSeq;

   typedef sequence <MOAttributeName> AttributeNameSet;

   /**

    * ModifyOperator defines the way in which an attribute value is to be

    * applied to an attribute in a modification of MO attributes.

    *

    * REPLACE: replace the current value with the provided value

    * ADD_VALUES: for a multi-valued attribute, add the provided values to the

    *   current list of values

    * REMOVE_VALUES: for a multi-valued attribute, remove the provided values

    *   from the current list of values

    * SET_TO_DEFAULT: set the attribute to its default value

    */

   enum ModifyOperator

   {

      REPLACE,

      ADD_VALUES,

      REMOVE_VALUES,

      SET_TO_DEFAULT

   };

   /**

    * AttributeModification defines an attribute value and the way it is to

    * be applied to an attribute in a modification of MO attributes.

    * It contains:

    * - the name of the attribute to modify ("name" field),

    * - the value to apply to this attribute ("value" field),

    * - the way the attribute value is to be applied to the attribute

    *   ("operator" field).

    */

   struct AttributeModification

   {

      MOAttributeName name;

      MOAttributeValue value;

      ModifyOperator operator;

   };

   typedef sequence <AttributeModification> AttributeModificationSet;

   typedef string ManagerReference;

   /* ManagerReferenceOpt is an optional type.

   If the discriminator is true the value is present.

   Otherwise the value is null.

   */

   union ManagerReferenceOpt switch (boolean)

   {

      case TRUE: ManagerReference value;

   };

};

#endif // _BASIC_CM_IRP_CONST_DEFS_IDL_

	Next Modified Section


A.2
IDL specification (file name "BasicCMIRPSystem.idl")

//File: BasicCMIRPSystem.idl

#ifndef _BASIC_CM_IRP_SYSTEM_IDL_

#define _BASIC_CM_IRP_SYSTEM_IDL_

#include <ManagedGenericIRPConstDefs.idl>
#include <ManagedGenericIRPSystem.idl>
#include <BasicCMIRPConstDefs.idl>
// This statement must appear after all include statements

#pragma prefix "3gppsa5.org"

module BasicCmIRPSystem

{

   exception IllegalFilterFormatException {

      string reason;

   };

   exception IllegalDNFormatException {

      string reason;

   };

   exception IllegalScopeTypeException {

      string reason;

   };

   exception IllegalScopeLevelException {

      string reason;

   };

   exception UndefinedMOException {

      string reason;

   };

   exception UndefinedScopeException {

      string reason;

   };

   exception FilterComplexityLimit {

      string reason;

   };

   exception DuplicateMO {};

   exception CreateNotAllowed {};

   exception ObjectClassMismatch {};

   exception NoSuchObjectClass {

      BasicCMIRPConstDefs::MOClass objectClass;

   };

   exception ParentObjectDoesNotExist {};

   /**

    * System otherwise fails to complete the operation. System can provide

    * reason to qualify the exception. The semantics carried in reason

    * is outside the scope of this IRP.

    */

   exception NextBasicCmInformations { string reason; };

   exception NextDeleteErrors { string reason; };

   exception NextModifyErrors { string reason; };

   exception DestroyException { string reason; };

   exception GetBasicCmIRPVersion { string reason; };

   exception GetBasicCmIRPOperationProfile { string reason; };

   exception GetBasicCmIRPNotificationProfile { string reason; };

   exception FindManagedObjects { string reason; };

   exception CreateManagedObject { string reason; };

   exception DeleteManagedObjects { string reason; };

   exception ModifyManagedObjects { string reason; };

   /**

   The BasicCmInformationIterator is used to iterate through a snapshot of

   Managed Object Information when IRPManager invokes find_managed_objects.

   IRPManager uses it to pace the return of Managed Object Information.

   IRPAgent controls the life-cycle of the iterator. However, a destroy

   operation is provided to handle the case where IRPManager wants to stop

   the iteration procedure before reaching the last iteration.

   */

   interface BasicCmInformationIterator

   {

      /**

      This method returns between 1 and "how_many" Managed Object information.

      The IRPAgent may return less than "how_many" items even if there are

      more items to return. "how_many" must be non-zero. Return TRUE if there

      may be more Managed Object information to return. Return FALSE if there

      are no more Managed Object information to be returned.

      If FALSE is returned, the IRPAgent will automatically destroy the

      iterator.

      @parm how_many how many elements to return in the "fetched_elements" out

       parameter.

      @parm fetched_elements the elements.

      @returns A boolean indicating if any elements are returned.

       "fetched_elements" is empty when the BasicCmInformationIterator is

       empty.

      */

      boolean next_basic_cm_informations (

         in unsigned short how_many,

         out BasicCMIRPConstDefs::ResultSet fetched_elements

      )

      raises (

         NextBasicCmInformations,

         ManagedGenericIRPSystem::InvalidParameter,

         ManagedGenericIRPSystem::OperationNotSupported);

      /**

      This method destroys the iterator.

      */

      void destroy ()

      raises (

         DestroyException,

         ManagedGenericIRPSystem::OperationNotSupported);

   }; // end of BasicCmInformationIterator

   /**

   The DeleteResultIterator is used to iterate through the list of deleted MOs

   when IRPManager invokes method "delete_managed_objects".

   IRPManager uses it to pace the return of Managed Object Information.

   IRPAgent controls the life-cycle of the iterator. However, a destroy

   operation is provided to handle the case where IRPManager wants to stop

   the iteration procedure before reaching the last iteration.

   */

   interface DeleteResultIterator : BasicCmInformationIterator

   {

      /**

      Inherited method "next_basic_cm_informations" has the same behaviour as

      for interface BasicCmInformationIterator, except that:

      - The Managed Object information returned in parameter

        "fetched_elements" contains only the DNs of the deleted MOs

        (no attributes are returned).

      - If FALSE is returned, the IRPAgent will not automatically destroy the

        iterator.

      */

      /**

      This method returns between 0 and "how_many" deletion errors. The

      IRPAgent may return less than "how_many" items even if there are more

      items to return. "how_many" must be non-zero. Return TRUE if there are

      more deletion errors to return. Return FALSE if there are no more

      deletion errors to be returned.

      If FALSE is returned and last call to inherited method

      "next_basic_cm_informations" also returned FALSE (i.e. no more Managed

      Object information to be returned), the IRPAgent will automatically

      destroy the iterator.

      @parm how_many: how many deletion errors to return in the

       "fetched_delete_errors" out parameter.

      @parm fetched_delete_errors: the deletion errors.

      @returns: a boolean indicating if any deletion errors are returned.

      */

      boolean next_delete_errors (

         in unsigned short how_many,

         out BasicCMIRPConstDefs::DeleteErrorSeq fetched_delete_errors

      )

      raises (

         NextDeleteErrors,

         ManagedGenericIRPSystem::InvalidParameter);

   }; // end of DeleteResultIterator

   /**

   The ModifyResultIterator is used to iterate through the list of modified

   MOs when IRPManager invokes method "modify_managed_objects".

   IRPManager uses it to pace the return of Managed Object Information.

   IRPAgent controls the life-cycle of the iterator. However, a destroy

   operation is provided to handle the case where IRPManager wants to stop

   the iteration procedure before reaching the last iteration.

   */

   interface ModifyResultIterator : BasicCmInformationIterator

   {

      /**

      Inherited method "next_basic_cm_informations" has the same behaviour as

      for interface BasicCmInformationIterator, except that:

      - The Managed Object information returned in parameter

        "fetched_elements" contains DNs and attributes of the modified MOs.

      - If FALSE is returned, the IRPAgent will not automatically destroy the

        iterator.

      */

      /**

      This method returns between 0 and "how_many" modification errors. The

      IRPAgent may return less than "how_many" items even if there are more

      items to return. "how_many" must be non-zero. Return TRUE if there are

      more modification errors to return. Return FALSE if there are no more

      modification errors to be returned.

      If FALSE is returned and last call to inherited method

      "next_basic_cm_informations" also returned FALSE (i.e. no more Managed

      Object information to be returned), the IRPAgent will automatically

      destroy the iterator.

      @parm how_many: how many modification errors to return in the

       "fetched_modify_errors" out parameter.

      @parm fetched_modify_errors: the modification errors.

      @returns: a boolean indicating if any modification errors are returned.

      */

      boolean next_modification_errors (

         in unsigned short how_many,

         out BasicCMIRPConstDefs::ModifyAttributeErrorsSeq

            fetched_modify_errors

      )

      raises (

         NextModifyErrors,

         ManagedGenericIRPSystem::InvalidParameter);

   }; // end of ModifyResultIterator

   /**

     * The BasicCmIrpOperations interface.

     * Supports a number of Resource Model versions.

     */

   interface BasicCmIrpOperations

   {

      /**

       * Get the version(s) of the interface

       *

       * @raises GetBasicCmIRPVersion when the system for some reason

       *   can not return the supported versions.

       * @returns all supported versions.

       */

      ManagedGenericIRPConstDefs::VersionNumberSet get_basic_cm_irp_version()

         raises (GetBasicCmIRPVersion);

      /**

       * Return the operation profile for a specific Basic CM IRP version.

       *

       * @raises GetBasicCmIRPOperationProfile when the system for some reason

       *   cannot return the supported operations and parameters.

       * @returns the list of all supported operations and their supported

       *   parameters for the specified version.

       */

      ManagedGenericIRPConstDefs::MethodList get_basic_cm_irp_operation_profile

      (

         in ManagedGenericIRPConstDefs::VersionNumber basic_cm_irp_version

      )

      raises (

         GetBasicCmIRPOperationProfile,

         ManagedGenericIRPSystem::OperationNotSupported,

         ManagedGenericIRPSystem::InvalidParameter);

      /**

       * Return the notification profile for a specific Basic CM IRP version.

       *

       * @raises GetBasicCmIRPNotificationProfile when the system for some

       *   reason cannot return the supported notifications and parameters.

       * @returns the list of all supported notifications and their supported

       *   parameters for the specified version.

       */

      ManagedGenericIRPConstDefs::MethodList

         get_basic_cm_irp_notification_profile (

            in ManagedGenericIRPConstDefs::VersionNumber basic_cm_irp_version

      )

      raises (

         GetBasicCmIRPNotificationProfile,

         ManagedGenericIRPSystem::OperationNotSupported,

         ManagedGenericIRPSystem::InvalidParameter);

      /**

       * Performs a containment search, using a SearchControl to

       * control the search and the returned results.

       *

       * All MOs in the scope constitute a set that the filter works on.

       * The result BasicCmInformationIterator contains all matched MOs,

       * with the amount of detail specified in the SearchControl.

       * For the special case when no managed objects are matched in

       * find_managed_objects, the BasicCmInformationIterator will be returned.

       * Executing the next_basicCmInformations in the

       * BasicCmInformationIterator will return FALSE for

       * completion.

       *

       * @parm base_object The start MO in the containment tree.

       * @parm search_control the SearchControl to use.

       * @parm requested_attributes defines which attributes to get.

       *   If this parameter is empty (""), all attributes shall

       *   be returned. In this version this is the only supported semantics.

       *   Note that this argument is only

       *   relevant if ResultContents in the search control is

       *   specifed to NAMES_AND_ATTRIBUTES.

       *

       * @raises ManagedGenericIRPSystem::ValueNotSupported if a valid but

       * unsupported parameter value is passed. E.g. the contents

       * field in the searchcontrol parameter contains the value NAMES and

       * the optional getContainment IS operation is not supported.

       * @raises UndefinedMOException The MO does not exist.

       * @raises IllegalDNFormatException The dn syntax string is

       * malformed.

       * @raises IllegalScopeTypeException The ScopeType in scope contains

       * an illegal value.

       * @raises IllegalScopeLevelException The scope level is negative

       * (<0).

       * @raises IllegalFilterFormatException The filter string is

       * malformed.

       * @raises FilterComplexityLimit if the filter syntax is correct,

       *   but the filter is too complex to be processed by the IRPAgent.

       * @see SearchControl

       * @see BasicCmInformationIterator

       */

      BasicCmInformationIterator find_managed_objects(

         in BasicCMIRPConstDefs::DN base_object,

         in BasicCMIRPConstDefs::SearchControl search_control,

         in BasicCMIRPConstDefs::AttributeNameSet requested_attributes,

         in BasicCMIRPConstDefs::ManagerReferenceOpt manager_reference)
      raises (

         FindManagedObjects,

         ManagedGenericIRPSystem::ParameterNotSupported,

         ManagedGenericIRPSystem::InvalidParameter,

         ManagedGenericIRPSystem::ValueNotSupported,

         ManagedGenericIRPSystem::OperationNotSupported,

         UndefinedMOException,

         IllegalDNFormatException,

         UndefinedScopeException,

         IllegalScopeTypeException,

         IllegalScopeLevelException,

         IllegalFilterFormatException,

         FilterComplexityLimit);

      /**

       * Performs the creation of a MO instance in the MIB maintained

       * by the IRPAgent.

       *

       * @parm object_name: the distinguished name of the MO to create.

       * @parm reference_object: the distinguished name of a reference MO.

       * @parm attributes: in input, initial attribute values for the MO to

       *   create; in output, actual attribute values of the created MO.

       * @parm attribute_errors: errors, related to attributes, that caused the

       *   creation of the MO to fail.

       *

       * @raises ManagedGenericIRPSystem::OperationNotSupported: The operation

       *   is not supported.

       * @raises ManagedGenericIRPSystem::ParameterNotSupported: An optional

       *   parameter is not supported.

       * @raises ManagedGenericIRPSystem::InvalidParameter: An invalid

       *   parameter value has been provided.

       * @raises UndefinedMOException: The MO does not exist.

       * @raises IllegalDNFormatException: The DN syntax string is malformed.

       * @raises DuplicateMO: A MO already exist with the same DN as the one

       *   to create.

       * @raises CreateNotAllowed: The creation of the MO is not allowed.

       * @raises ObjectClassMismatch: The object class of the MO to create does

       *   not match with the object class of the provided reference MO.

       * @raises NoSuchObjectClass: The class of the object to create is not

       *   recognized.

       * @raises ParentObjectDoesNotExist: The parent MO instance of the

       *  ManagedEntity specified to be created does not exist.

       */

      void create_managed_object (

         in BasicCMIRPConstDefs::DN object_name,

         in BasicCMIRPConstDefs::DN reference_object,

         inout BasicCMIRPConstDefs::MOAttributeSet attributes,

         out BasicCMIRPConstDefs::AttributeErrorSeq attribute_errors

      )

      raises (

         CreateManagedObject,

         ManagedGenericIRPSystem::OperationNotSupported,

         ManagedGenericIRPSystem::ParameterNotSupported,

         ManagedGenericIRPSystem::InvalidParameter,

         UndefinedMOException,

         IllegalDNFormatException,

         DuplicateMO,

         CreateNotAllowed,

         ObjectClassMismatch,

         NoSuchObjectClass,

         ParentObjectDoesNotExist);

      /**

       * Performs the deletion of one or more MO instances from the MIB

       * maintained by the IRPAgent, using a SearchControl to control the

       * instances to be deleted.

       *

       * All MOs in the scope constitute a set that the filter works on.

       * All matched MOs will be deleted by this operation.

       * The returned DeleteResultIterator is used to retrieve the DNs of the

       * MOs deleted and the errors that may have occurred preventing deletion

       * of some MOs.

       * For the special case when no managed objects are matched in

       * delete_managed_objects, the DeleteResultIterator will be returned.

       * Executing the next_basicCmInformations in the DeleteResultIterator

       * will return FALSE for completion.

       *

       * @parm base_object: the start MO in the containment tree.

       * @parm search_control: the SearchControl to use; field "contents" has no

       *   meaning here and shall be ignored.

       * @returns: a DeleteResultIterator (see above).

       *

       * @raises ManagedGenericIRPSystem::OperationNotSupported: The operation

       *   is not supported.

       * @raises ManagedGenericIRPSystem::InvalidParameter: An invalid

       *   parameter value has been provided.

       * @raises UndefinedMOException: The MO does not exist.

       * @raises IllegalDNFormatException: The DN syntax string is malformed.

       * @raises IllegalScopeTypeException: The ScopeType in scope contains

       *   an illegal value.

       * @raises IllegalScopeLevelException: The scope level is negative (<0).

       * @raises IllegalFilterFormatException: The filter string is malformed.

       * @raises FilterComplexityLimit: The filter syntax is correct,

       *   but the filter is too complex to be processed by the IRPAgent.

       */

      DeleteResultIterator delete_managed_objects (

         in BasicCMIRPConstDefs::DN base_object,

         in BasicCMIRPConstDefs::SearchControl search_control

      )

      raises (

         DeleteManagedObjects,

         ManagedGenericIRPSystem::OperationNotSupported,

         ManagedGenericIRPSystem::InvalidParameter,

         UndefinedMOException,

         IllegalDNFormatException,

         UndefinedScopeException,

         IllegalScopeTypeException,

         IllegalScopeLevelException,

         IllegalFilterFormatException,

         FilterComplexityLimit);

      /**

       * Performs the modification of MO attributes. One or more MOs attributes

       * may be modified according to a SearchControl.

       *

       * All MOs in the scope constitute a set that the filter works on.

       * All matched MOs will have their attributes modified by this operation.

       * The returned ModifyResultIterator is used to retrieve the DNs of the

       * modified MOs together with the values of the modified attributes, and

       * the errors that may have occurred preventing modification of some

       * attributes.

       * For the special case when no managed objects are matched in

       * modify_managed_objects, the ModifyResultIterator will be returned.

       * Executing the next_basicCmInformations in the ModifyResultIterator

       * will return FALSE for completion.

       *

       * @parm base_object: the start MO in the containment tree.

       * @parm search_control: the SearchControl to use; field "contents" has no

       *   meaning here and shall be ignored.

       * @parm modifications: the values for the attributes to modify and

       *   the way those values are to be applied to the attributes.

       * @returns: a ModifyResultIterator (see above).

       *

       * @raises ManagedGenericIRPSystem::OperationNotSupported: The operation

       *   is not supported

       * @raises ManagedGenericIRPSystem::InvalidParameter: An invalid

       *   parameter value has been provided

       * @raises UndefinedMOException: The MO does not exist.

       * @raises IllegalDNFormatException: The DN syntax string is malformed.

       * @raises IllegalScopeTypeException: The ScopeType in scope contains

       *   an illegal value.

       * @raises IllegalScopeLevelException: The scope level is negative (<0).

       * @raises IllegalFilterFormatException: The filter string is malformed.

       * @raises FilterComplexityLimit: The filter syntax is correct,

       *   but the filter is too complex to be processed by the IRPAgent.

       */

      ModifyResultIterator modify_managed_objects (

         in BasicCMIRPConstDefs::DN base_object,

         in BasicCMIRPConstDefs::SearchControl search_control,

         in BasicCMIRPConstDefs::AttributeModificationSet modifications

      )

      raises (

         ModifyManagedObjects,

         ManagedGenericIRPSystem::OperationNotSupported,

         ManagedGenericIRPSystem::InvalidParameter,

         UndefinedMOException,

         IllegalDNFormatException,

         UndefinedScopeException,

         IllegalScopeTypeException,

         IllegalScopeLevelException,

         IllegalFilterFormatException,

         FilterComplexityLimit);

   };

};

#endif // _BASIC_CM_IRP_SYSTEM_IDL_

	End of modifications

